
1 課題文
2次元非圧縮ポテンシャル流の定常流で、流体の占める領域 Ω と、その境界 Γ = ∂Ω での

流速の法線成分 Vn := v ·n が分かっている場合に、速度ポテンシャル ϕ を計算して、等ポテ
ンシャル線と速度場を可視化せよ。
領域Ωとしては、円盤領域以外のものを選ぶこと (多角形領域は比較的簡単でお勧め。楕円
領域を考える人がいると思われるが、楕円に沿っての積分は難しいことが多いのであまり勧め
ない。)。境界値 (流速の法線成分) Vn も自分で興味のあるもの、自分の都合の良いものを選
んで良い (後の注意を読んでおくこと)。
以下でサンプル・プログラムを紹介する。それを理解した上で、領域 Ω と境界データ Vn を

自分が決めたものに書き換えれば良い。自由度は高いので、工夫・遊び心発揮を期待する。
出来る限り、流線も可視化せよ。流線の書き方には色々なやり方があるが、流れ関数を求め
るやり方を勧めておく (それを求めるにもポテンシャル問題が使えたりするので)。

2 参考情報など
• FreeFem++ の使い方については、

– FreeFEM-documentation.pdf (公式ドキュメント)

https://doc.freefem.org/pdf/FreeFEM-documentation.pdf

(本当は Mac の中にあるだけれど、FreeFem++ のバージョンによって場所が違う
ので…1)

– 「FreeFem++ の紹介」桂田が書いた紹介文書
https://m-katsurada.sakura.ne.jp/labo/text/welcome-to-freefem/

– 「FreeFem++ノート」桂田が書いた文法メモ
https://m-katsurada.sakura.ne.jp/labo/text/freefem-note/

• サンプル・プログラムを用意してある。入手法をまとめて示しておく。
ターミナルで実行する� �

curl -O https://m-katsurada.sakura.ne.jp/program/fem/poisson-kikuchi.edp

curl -O https://m-katsurada.sakura.ne.jp/ana/potential2d-v-2024.edp

curl -O https://m-katsurada.sakura.ne.jp/ana/sample0.edp

curl -O https://m-katsurada.sakura.ne.jp/ana/sample1.edp

curl -O https://m-katsurada.sakura.ne.jp/ana/sample2.edp� �
poisson-kikuchi.edp は、正方形領域におけるPoisson方程式の境界値問題を解くプ

ログラムで、すでに紹介済み。これについては、「FreeFem++ で菊地 [1] の Poisson 方
程式の例題を解く」を見るとよい。多角形領域の扱い方が分かるはず。
それ以外の potential2d-v-2024.edp, sample0.edp, sample1.edp, sample2.edpは、

円盤領域で

△ϕ = 0 (in Ω)(1a)

∂ϕ

∂n
= Vn (on ∂Ω)(1b)

を解くプログラムである。中身は境界条件の設定の仕方が違うだけで他はほとんど同じ。

弱形式は、すでに説明したように∫∫
Ω

(ϕxvx + ϕyvy) dx dy =

∫
∂Ω

Vnv dσ (任意の試験関数 v に対して).

この左辺は FreeFem++ のプログラム中では、int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v))

のようにすれば良い。右辺は少し工夫が必要である (FreeFem++ の機能が今ひとつなため)。

3 注意 (必ず読むこと)

次が非常に重要である。
Vn は

∫
∂Ω

Vn dσ = 0 を満たしている必要がある。実際、∂ϕ
∂n
の定義と Gaussの発散定理から∫

∂Ω

Vn dσ =

∫
∂Ω

∂ϕ

∂n
dσ =

∫
∂Ω

gradϕ · n dσ =

∫
Ω

div gradϕ dx =

∫
Ω

△ϕ dx =

∫
Ω

0 dx = 0.

1一応 find /Applications/FreeFem++.app -name FreeFEM-documentation.pdf -print で表示されま
すが…

非圧縮流 (div v = 0) の速度場 v が分かっている場合に、Vn = v ·n と置いた場合は、この
条件は必ず成り立っている (

∫
∂Ω

Vn dσ =

∫
∂Ω

v · n dσ =

∫
Ω

div v dx =

∫
Ω

0 dx = 0)。

そうでない場合には、
∫
∂Ω

Vn dσ = 0 が成り立つように、注意深く Vn を選ぶ必要がある。

4 FAQ (よくされる質問)

「境界値 Vn を場合分けを含む関数としたいが、FreeFem++ がプログラムを受け付けてく
れません。どうすればいいですか？」
(FreeFem++ ってしょうがないですね…) 例えば以下の 2つの解決策があります。

(a) 弱形式には複数の int1d() が指定できるので、境界を分割して、その部分ごとに (場合分
けを含まない)境界値を与えるようにプログラムを書く。

(b) 例えば (x>1 && y>0) のような式は、条件が成り立つならば 1, 成り立たないならば 0と
いう値を持つので、ifを使わずに、式だけで場合分けを含む関数が記述できる。
次のサンプル・プログラムでは、(b) を用いているが、ある程度以上複雑になった場合は、
私のおすすめは (a) である。付録の sample1.edp を見て下さい。

5 サンプルプログラム potential2d-v-2024.edp 解説
potential2d-v-2024.edp� �

1 // potential2d-v-2024.edp

2 // curl -O https://m-katsurada.sakura.ne.jp/complex2/potential2d-v-2024.edp

3 // 2次元非圧縮ポテンシャル流
4 // 速度ポテンシャル，速度を求め、等ポテンシャル線, 速度場を描く
5

6 border Gamma(t=0,2*pi) { x = cos(t); y = sin(t); } // 円盤領域
7 int m=40;

8 mesh Th=buildmesh(Gamma(m));

9 plot(Th, wait=1, ps="Th.eps");

10 // 次の 2行は区分 1次多項式を使うという意味
11 fespace Vh(Th,P1);

12 Vh phi, v;

13 // 境界条件の設定 右上と左下のみ Vn=x+2y (これは確かに ∫ Vn dσ=0を満たす)

14 func Vn=((x>0&&y>0) || (x<0&&y<0))*(x+2*y);

15

16 // 速度ポテンシャルφを求める
17 solve Laplace(phi,v) =

18 int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v)) -int1d(Th,Gamma)(Vn*v);

19

20 // 平均を 0にする (解の一意性がないため、時々生じる大きなズレを消す
21 real meanphi = int2d(Th)(phi)/Th.area; // Th.area は int2d(Th)(1); でも計算可能
22 phi = phi - meanphi;

23

24 // φの等高線 (等ポテンシャル線) を描く
25 plot(phi,ps="contourpotential.eps",wait=1);

26

27 // ベクトル場 (v1,v2)=∇φ を描く (ちょっと雑なやり方)

28 Vh v1, v2;

29 v1=dx(phi); v2=dy(phi);

30 plot([v1,v2],ps="vectorfield.eps",wait=1);

31

32 // 等ポテンシャル線とベクトル場を同時に描く
33 plot([v1,v2],phi,ps="both.eps", wait=1);� �

• 6行目で領域Ωの境界 Γ := ∂Ωを指定している (それで Ω が決まる)。

• 8行目で、Γをm分割して、Γの囲む範囲を三角形分割して、それを mesh 型の変数 Th

に代入している。

• 11行目、有限要素空間 Vh を区分的 1次関数の空間と定義している。この辺については
この講義では説明を省略する (知りたい人は有限要素法のテキストを読んで下さい)。

• 12行目、ϕ と試験関数 v を Vh の要素とする。

• 14行目、境界値 Vn の設定をしている。ここでは

Vn(x, y) =

{
x+ 2y ((x > 0かつ y > 0)または (x < 0かつ y < 0)のとき)

0 (それ以外)

としている。(後の例 sample0.edp では Vn(x, y) = x+2y =

(
1

2

)
·n としていて、これ

は v =

(
1

2

)
という一様流となる。) これは確かに

∫
∂Ω

Vn dσ = 0 を満たしている。

• 17～18行目、弱形式の定義。

• 21～22行目、Neumann境界値問題は定数だけの不定性を持つため (ϕが解ならば ϕ+定数

も解)、時々大きなズレが生じ、解の描画などで問題が生じる。平均値

∫∫
Ω

ϕ(x, y)dx dy∫∫
Ω

dx dy

を引くことで、平均= 0にしておく。

6 流線を描くために: 流れ関数 ψを求める
流線があると流れの様子がよく分かるので、流線を描く価値は高い。
2次元の場合は、流れ関数の等高線として流線がかける。流れ関数 ψ は、線積分で表すこ
とができる:

ψ(x) =

∫
Cx

ψx dx+ ψy dy =

∫
Cx

(
−v
u

)
· dr =

∫
Cx

(
−v
u

)
· t ds (x ∈ Ω)

ここで Cx は定点 a と x を結ぶ曲線である。
領域が Jordan領域 (1つの単純閉曲線で囲まれた領域) であれば、x が境界上の点である場
合、定点 a と Cx を境界上に選ぶことで、(

(
−v
u

)
· t = v · n. また境界上での v · n = Vn は

知っていると仮定しているので)

ψ(x) =

∫
Cx

v · n ds =

∫
Cx

Vn ds (x ∈ ∂Ω).

これから、境界上の点 x における ψ の値 g(x) := ψ(x) (x ∈ ∂Ω) が得られる。それを用いて

△ψ = 0 (in Ω), ψ(x) = g(x) (on ∂Ω)

という Laplace 方程式の Dirichlet 境界値問題を解くことで流れ関数 ψ が得られる。
Dirichlet境界条件のある場合のプログラムは、上記の poisson-kikuchi.edpが参考になる。

A サンプル・プログラム解説
以下すべて単位円版領域において

△ϕ = 0 in Ω,
∂ϕ

∂n
= Vn on ∂Ω

を解くプログラムである。境界条件の指定の部分が違っている。

A.1 一様流のプログラム sample0.edp

サンプルプログラム sample0.edpでは、単位円盤領域Ω = {(x, y) ∈ R2 | x2 + y2 < 1} にお
ける一様流 v =

(
1

2

)
の場合のプログラムである。単位円盤なので n =

(
x

y

)
(ここは良く考

えること。例えば楕円の場合は n はかなり複雑な式になる。). これから

Vn = v · n =

(
1

2

)
·

(
x

y

)
= x+ 2y.

プログラムでは、func Vn=x+2*y; で Vnを定義して、弱形式中で int1d(Th,Gamma)(Vn*v)

と使っている。
(v が Ω で定数関数なので) div v = 0 であるから、当然

∫
Γ

Vn dσ = 0 も成り立つ。
sample0.edp� �

// sample0.edp

// https://m-katsurada.sakura.ne.jp/complex2/sample0.edp

// 2次元非圧縮ポテンシャル流
// 速度ポテンシャル，速度を求め、等ポテンシャル線, 速度場を描く

border Gamma(t=0,2*pi) { x = cos(t); y = sin(t); } // 円盤領域
int m=60;

mesh Th=buildmesh(Gamma(m));

plot(Th, wait=1, ps="Th.eps");

// 次の 2行は区分 1次多項式を使うという意味
fespace Vh(Th,P1);

Vh phi, v, v1, v2;

// 境界条件の設定
func Vn=x+2*y;// Ωが単位円で, V=(1,2) のとき V・n=x+2y

// 速度ポテンシャルφを求め、その等高線 (等ポテンシャル線) を描く
solve Laplace(phi,v) =

int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v)) -int1d(Th,Gamma)(Vn*v);

real meanphi=int2d(Th)(phi)/Th.area;

phi=phi-meanphi;

plot(phi,ps="contourpotential.eps",wait=1);

// ベクトル場 (v1,v2)=∇φ を描く (ちょっと雑なやり方)

v1=dx(phi); v2=dy(phi);

plot([v1,v2],ps="vectorfield.eps",wait=1);

// 等ポテンシャル線とベクトル場を同時に描く
plot([v1,v2],phi,ps="both.eps", wait=1);� �

図 1: 一様流 (sample0.edp)

A.2 sample1.edp, sample2.edp

単位円盤領域の境界である単位円周上で、0 ≤ θ ≤ π/6, π/6 ≤ θ ≤ π, π ≤ θ ≤ 4π/3

4π/3 ≤ θ ≤ 2π の範囲をそれぞれ Γ1, Γ2, Γ3, Γ4 とする。すなわち

Γ1 = {(cos θ, sin θ) | 0 ≤ θ ≤ π/6} ,
Γ2 = {(cos θ, sin θ) | π/6 ≤ θ ≤ π} ,
Γ3 = {(cos θ, sin θ) | π ≤ θ ≤ 4π/3} ,
Γ4 = {(cos θ, sin θ) | 4π/3 ≤ θ ≤ 2π} .

境界条件 ∂ϕ
∂n

= Vn における Vn は、次のようになっているとする。

Vn =


−2 (on Γ1)

0 (on Γ2)

1 (on Γ3)

0 (on Γ4).

念のため
∫
∂Ω

Vn dσ = 0 の確認� �
∫
γ

dσ = γ の長さ に注意すると
∫
∂Ω

Vn dσ =

∫
Γ1

Vn dσ +

∫
Γ2

Vn dσ +

∫
Γ3

Vn dσ +

∫
Γ4

Vn dσ

= −2

∫
Γ1

dσ + 0 + 1 ·
∫
Γ3

dσ + 0 = −2 · π
6
+ 1 · π

3
= 0.

� �
sample1.edp, sample2.edp はともにこの問題を解くプログラムである。

もしも、プログラム中で Vn がうまく定義できれば、sample0.edp と同様に� �
solve Laplace(phi,v) =

int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v)) -int1d(Th,Gamma)(Vn*v);� �
というコードが使える。

しかし Vn を 1行の関数として定義するのはあまり簡単でない。一応出来なくはなくて、そ
れは後で紹介するが (sample2.edp)、ここでは次のようにすることを勧める。
要点は ∫

∂Ω

Vnv dσ =
4∑

j=1

∫
Γj

Vnv dσ =

∫
Γ1

Vnv dσ +

∫
Γ3

Vnv dσ

と分解することである (Γ2, Γ4 では Vn = 0 であるから、
∫
Γj

Vnv dσ (j = 2, 4) は必要ない)。
Γ1, Γ3 における Vn を、それぞれ Vn1, Vn3 と定義すれば� �
solve Laplace(phi,v) =

int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v))

-int1d(Th,Gamma1)(Vn1*v)-int1d(Th,Gamma3)(Vn3*v);� �
とすれば良い 。このためには、もちろんプログラム中で Γ1, Γ2, Γ3, Γ4 を用意する必要がある
が、それは難しくない (境界を 4つに分けた正方形領域のプログラム poisson-kikuchi.edp

と同様のことをすれば良い)。以上の考察から、次のプログラムが得られる。
sample1.edp� �

// sample1.edp

// https://m-katsurada.sakura.ne.jp/complex2/sample1.edp

// 2次元非圧縮ポテンシャル流
// 速度ポテンシャル，速度を求め、等ポテンシャル線, 速度場を描く

border Gamma1(t=0,pi/6) { x = cos(t); y = sin(t); }

border Gamma2(t=pi/6,pi) { x = cos(t); y = sin(t); }

border Gamma3(t=pi,4*pi/3) { x = cos(t); y = sin(t); }

border Gamma4(t=4*pi/3,2*pi) { x = cos(t); y = sin(t); }

int m=5;

mesh Th=buildmesh(Gamma1(m)+Gamma2(5*m)+Gamma3(2*m)+Gamma4(4*m));

plot(Th, wait=1, ps="Th.eps");

// 次の 2行は区分 1次多項式を使うという意味
fespace Vh(Th,P1);

Vh phi, v;

// Gamma1, Gamma3 上での Vn

func Vn1 = -2.0;

func Vn3 = 1.0;

// 速度ポテンシャルφを求め、その等高線 (等ポテンシャル線) を描く
solve Laplace(phi,v) =

int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v))

-int1d(Th,Gamma1)(Vn1*v)-int1d(Th,Gamma3)(Vn3*v);

real mean=int2d(Th)(phi)/Th.area; // Th.area は int2d(Th)(1.0)

phi=phi-mean;

plot(phi,ps="contourpotential.eps",wait=1);

// ベクトル場 (v1,v2)=∇φ を描く (ちょっと雑なやり方)

Vh v1, v2;

v1=dx(phi); v2=dy(phi);

plot([v1,v2],ps="vectorfield.eps",wait=1);

// 等ポテンシャル線とベクトル場を同時に描く
plot([v1,v2],phi,ps="both.eps", wait=1);� �
以下、参考までに、Vnを1命令 (長いので2行になっている)で書いたプログラムsample2.edp

を紹介しておく。いわゆる偏角の主値 (値が (−π, π] に属する) を返す関数 atan2(y,x) を使

図 2: sample1.edp の結果 (θ ∈ [0, π/6]で勢い良く入り、θ ∈ [π, 4π/3] でゆっくり出て行く)

うが、やや不自然な感じがするのは否めない (角度の範囲が [0, 2π) でないことにも注意が必
要である)。

強引に 1行関数を作る� �
func Vn=((atan2(y,x)>=0.0&&atan2(y,x)<=pi/6) * (-2.0)

+ (atan2(y,x)>=-pi&&atan2(y,x)<=-2.0*pi/3.0) * (1.0));� �
sample2.edp� �

// sample2.edp

// https://m-katsurada.sakura.ne.jp/complex2/sample2.edp

// 2次元非圧縮ポテンシャル流
// 速度ポテンシャル，速度を求め、等ポテンシャル線, 速度場を描く

border Gamma(t=0,2*pi) { x = cos(t); y = sin(t); } // 円盤領域
int m=60;

mesh Th=buildmesh(Gamma(m));

plot(Th, wait=1, ps="Th.eps");

// 次の 2行は区分 1次多項式を使うという意味
fespace Vh(Th,P1);

Vh phi, v, v1, v2;

// 境界条件の設定 (atan2(y,x) は [-pi,pi) の範囲の値を返す)

func Vn=((atan2(y,x)>=0.0&&atan2(y,x)<=pi/6) * (-2.0)

+ (atan2(y,x)>=-pi&&atan2(y,x)<=-2.0*pi/3.0) * (1.0));

// 速度ポテンシャルφを求め、その等高線 (等ポテンシャル線) を描く
solve Laplace(phi,v) =

int2d(Th)(dx(phi)*dx(v)+dy(phi)*dy(v)) -int1d(Th,Gamma)(Vn*v);

real meanphi=int2d(Th)(phi)/Th.area;

phi=phi-meanphi;

plot(phi,ps="contourpotential.eps",wait=1);

// ベクトル場 (v1,v2)=∇φ を描く (ちょっと雑なやり方)

v1=dx(phi); v2=dy(phi);

plot([v1,v2],ps="vectorfield.eps",wait=1);

// 等ポテンシャル線とベクトル場を同時に描く
plot([v1,v2],phi,ps="both.eps", wait=1);� �

図 3: sample2.edp の結果 (θ ∈ [0, π/6]で勢い良く入り、θ ∈ [π, 4π/3] でゆっくり出て行く)

